

Data Restructuring Using SPSS

Danielle Popp, Tessa West, and David A. Kenny

These are our suggestions for data restructuring. If you know of any other strategies, please email us. Text in purple is SPSS syntax. All the syntax on this page is contained in the file named [datarestructuring.sps](#) which can be downloaded.

Sample Individual Data from Acitelli Data Set:

ID	Dyad	Partnum	Gender	Self1	Self2	Self3	Self4
1	3	1	1.00	4	4	3	4
2	3	2	-1.00	5	5	5	5
3	10	1	1.00	3	4	4	5
4	10	2	-1.00	4	5	4	5
5	11	1	1.00	4	3	5	5
6	11	2	-1.00	5	5	5	5
7	17	1	1.00	3	3	4	5
8	17	2	-1.00	4	4	4	4
9	21	1	1.00	4	4	5	5
10	21	2	-1.00	3	5	5	4

Note that the data already have Dyad and Partnum variables. (Make sure Partnum has no decimal places.) Partnum does not have to be ones and twos, but might be W and H, a “string” variable. It is advisable to create dummy cases if one of the two partners is missing. This data file is called [sampledata.sav](#) and can be downloaded.

To Restructure from Individual to Dyad:

First, you sort cases by dyad number and partner number.

```
SORT CASES BY
  dyad (A) partnum (A) .
```

Next, you will create a data file that will have both Partner 1’s and Partner 2’s data on the same line.

```
CASESTOVARS
  /id=dyad /index=partnum
  /groupby=index /autofix=no.
```

The resulting data set will have all of Partner 1’s data first and then all of Partner 2’s data. Partner 1’s data will be indicated by the original variable name (e.g., self1)

followed by 1 (self1.1); Partner 2's data will be indicated by the original variable name followed by 2 (self1.2) as below:

dyad	gender.1	self1.1	self2.1	self3.1	self4.1	gender.2	self1.2	self2.2	self3.2	self4.2
3	1	4	4	3	4	-1	5	5	5	5
10	1	3	4	4	5	-1	4	5	4	5
11	1	4	3	5	5	-1	5	5	5	5
17	1	3	3	4	5	-1	4	4	4	4
21	1	4	4	5	5	-1	3	5	5	4

Note there are two gender variables that are the same. One could be deleted. Note also the cases are ordered in numeric (1's before 2's) or alphabetic order (H's before W's).

To Restructure from Individual to Pairwise:

Follow instructions above to create the first dyad level data file. Note: The first data file will have Person 1's data on the left and Person 2's data on the right. Make sure to save the above data file under a different name (e.g., DyadA.sav).

Next, create a second dyad data set with Person 2's data on the left and Person 1's data on the right.

First, you must re-open the individual level data set started with. Next, you will recode the PARTNUM variable switching which member of the dyad is Person 1 and which is Person 2.

```
RECODE
  partnum (1=2) (2=1) .
EXECUTE .
```

The second line would be `partnum (H=W) (W=H)` if you used H and W. Next, you re-run the syntax from step 1.

```
SORT CASES BY
  dyad (A) partnum (A) .

CASESTOVARS
  /id=dyad /index=partnum
  /groupby=index /autofix=no.
```

You now have a second data set with Person 2's data on the left and Person 1's data on the right. Save this data set under a different file name (e.g., C:\DyadB.sav).

Next, open the first dyad data file (DyadA.sav). Now merge the files together to create the pairwise data set.

```
ADD FILES /FILE=*
  /FILE='C:\DyadB.sav'.
EXECUTE.
```

assuming that DyadB.sav is saved on the C drive. To check that you have reformatted the file correctly, sort the merged file by dyad.

SORT CASES BY
dyad (A) .

You should see two lines for each dyad, the first line will have Person 1's data first and Person 2's data second. The second line will have Person 2's data first and Person 1's data second.

Save your pairwise data file:

dyad	person	gender.1	self1.1	self2.1	self3.1	self4.1	gender.2	self1.2	self2.2	self3.2	self4.2
3	1	1	4	4	3	4	-1	5	5	5	5
3	2	-1	5	5	5	5	1	4	4	3	4
10	1	1	3	4	4	5	-1	4	5	4	5
10	2	-1	4	5	4	5	1	3	4	4	5
11	1	1	4	3	5	5	-1	5	5	5	5
11	2	-1	5	5	5	5	1	4	3	5	5
17	1	1	3	3	4	5	-1	4	4	4	4
17	2	-1	4	4	4	4	1	3	3	4	5
21	1	1	4	4	5	5	-1	3	5	5	4
21	2	-1	3	5	5	4	1	4	4	5	5

Note that "1" refers to the respondent and "2" to the partner.

To Restructure from Dyad to Pairwise:

As will be seen this is fairly complicated. If one has an Individual file, it would be advisable to convert that file to a Pairwise file. To restructure a Dyad file to pairwise file, first open your dyad level data file. For this example, we will use DyadA.sav.

First we will create a second copy of our data file that will be a working copy so that we do not write over the original data file.

`save outfile='c:\workingdata.sav'.`

Next, we will create a data that will contain only the data for the first person in every row – essentially, this will look like an individual level data file with the distinction being that these variables are uniquely named to be the first person in each row (e.g., aself1). We will save this data set with a unique name (e.g., actor.sav).

```
varstocases
/make agender from gender.1 gender.2
/make aself1 from self1.1 self1.2
/make aself2 from self2.1 self2.2
/make aself3 from self3.1 self3.2
/make aself4 from self4.1 self4.2
/index=person (2) /keep=dyad.
```

```
save outfile='c:\actor.sav'.
```

Next, we reopen the working data file and create a second data set with only the variables for the second person on each line. Again, these variables are uniquely named indicating they are for the second person on each line because we will be merging the two data sets together to create a final pairwise data set.

```
get file='c:\workingdata.sav'.
varstocases
/make pgender from gender.2 gender.1
/make pself1 from self1.2 self1.1
/make pself2 from self2.2 self2.1
/make pself3 from self3.2 self3.1
/make pself4 from self4.2 self4.1
/index=person (2) /keep=dyad.
```

Finally, we merge the two files together to create the pairwise data set.

```
match files file='c:\actor.sav'
/file=* /by dyad person.
execute.
```

dyad	person	agender	aself1	aself2	aself3	aself4	pgender	pself1	pself2	pself3	pself4
3	1	1	4	4	3	4	-1	5	5	5	5
3	2	-1	5	5	5	5	1	4	4	3	4
10	1	1	3	4	4	5	-1	4	5	4	5
10	2	-1	4	5	4	5	1	3	4	4	5
11	1	1	4	3	5	5	-1	5	5	5	5
11	2	-1	5	5	5	5	1	4	3	5	5
17	1	1	3	3	4	5	-1	4	4	4	4
17	2	-1	4	4	4	4	1	3	3	4	5
21	1	1	4	4	5	5	-1	3	5	5	4
21	2	-1	3	5	5	4	1	4	4	5	5

Note “a” variables refer to the respondent and “p” variables to the partner.